Supplementary Model Code 1: Pharmacokinetic model code

```
$PROBLEM    OXY 1 comp PK

$INPUT      C ID TIME DV GENDER WT GENOTYPE HT RACE SUA SCR BLQFL EVID
        AMT CMTT MDV

$DATA      Oxypurinol_NONMEM_file_110119.csv IGNORE=@

$SUBROUTINE ADVAN6 TRANS1 TOL=5

$MODEL      NCOMP=2             ; number of compartments
        COMP=(ABS,DEFDOSE)            ; absorption compartment
        COMP=(CENTRAL,DEFOBS)  ; plasma central compartment

;ng dose units
;hr time units
;ng/mL oxy concentration units

$PK

;-----------------PLASMA PK PARAMETERS----------------------

IF(GENDER.EQ.0) VGENDER = 1                    ; [V~GENDER]
IF(GENDER.EQ.1) VGENDER = ( 1 + THETA(6))
VCOV=VGENDER

TVCL = THETA(3)                          ; typical value of clearance
CL = THETA(3) * EXP(ETA(1))                    ; clearance

TVV = THETA(4)                          ; typical value of volume
TVV = VCOV*TVV                           ; [V~GENDER]
V  = TVV * EXP(ETA(2))                    ; volume

TVKA=THETA(5)                           ; typical value of formation rate constant
KA = TVKA * EXP(ETA(3))                      ; formation rate constant

KE = CL/V                              ; elimination rate constant from central compartment
F1 = 1                                 ; oral bioavailability

;-----------------ODES----------------------

$DES
DADT(1) = -KA * A(1)                       ; ODE for absorption compartment
```

```
DADT(2) = KA * A(1) - KE * A(2)                ; ODE for plasma central compartment

;-----------------PREDICTION & ERROR MODEL----------------------

$ERROR
IPRED=A(2)/V                                   ; prediction for PK units: ng/mL
WA=THETA(1)                                     ; additive error for PK
WP=THETA(2)                                     ; proportional error for PK

W = SQRT(WA**2+(WP*IPRED)**2) ; combined error model
IRES=DV-IPRED
IWRES=IRES/W
Y = IPRED + W*EPS(1)

;-----------------INITIAL ESTIMATES----------------------

$THETA
 0.01     FIX                                  ; additive error for PK
 0.253467                                       ; proportional error for PK
 1742.83                                        ; CL (mL/hr)
 57013.5                                        ; V (mL)
 0.77117                                        ; KA (1/hr)
 -0.248189                                      ; [V~GENDER]

$OMEGA
 0.0556119    ; ETA1 (CL)
 0.034849     ; ETA2 (V)
 0.304355     ; ETA3 (KA)

$SIGMA
 1  FIX

$ESTIMATION METHOD=1 INTERACTION PRINT=5 SIG=3 MAXEVAL=9999 NOABORT
$COVARIANCE UNCONDITIONAL

$TABLE      ID TIME IPRED DV GENDER GENOTYPE SCR WRES CWRES EVID CMTT MDV
        FILE=sdtab9
        NOPRINT ONEHEADER
```

Supplementary Model Code 2: Pharmacokinetic-pharmacodynamic model code

```
$PROBLEM    OXY 1 comp PK PD

$INPUT      C ID TIME DV GENDER WT GENOTYPE HT RACE SUA SCR BLQFL EVID
            AMT CMTT MDV FLAG SCRSUB

$DATA       OXY_PK_PD_022720.csv IGNORE=@

$SUBROUTINE ADVAN6 TRANS1 TOL=5

$MODEL      NCOMP=2              ; number of compartments
            COMP=(ABS,DEFDOSE)           ; absorption compartment
            COMP=(CENTRAL,DEFOBS)  ; plasma central compartment

;ng dose units
;hr time units
;ng/mL oxy concentration units
;mg/dL SUA units

$PK

;-----------------PLASMA PK PARAMETERS----------------------

IF(GENDER.EQ.0) VGENDER = 1                  ; [V~GENDER]
IF(GENDER.EQ.1) VGENDER = ( 1 + THETA(6))
VCOV=VGENDER

TVCL = THETA(3)                        ; typical value of clearance
CL = THETA(3) * EXP(ETA(1))                  ; clearance

TVV = THETA(4)                        ; typical value of volume
TVV = VCOV*TVV                          ; [V~GENDER]
V  = TVV * EXP(ETA(2))                   ; volume

TVKA=THETA(5)                          ; typical value of formation rate constant
KA = TVKA * EXP(ETA(3))                    ; formation rate constant

KE = CL/V                            ; elimination rate constant from central compartment
F1 = 1                              ; oral bioavailability

;-----------------PKPD PARAMETERS----------------------

BASESCR = ( 1 + THETA(14)*(SCR - 0.78))          ; [BASE~SCR]
```

```
IF(GENOTYPE.EQ.0) BASEGENOTYPE = 1              ; [BASE~GENOTYPE]
IF(GENOTYPE.EQ.1) BASEGENOTYPE = ( 1 + THETA(12))
IF(GENOTYPE.EQ.2) BASEGENOTYPE = ( 1 + THETA(13))

BASECOV=BASEGENOTYPE*BASESCR

TVEMAX = THETA(7)                          ; typical value of Emax
EMAX = TVEMAX * EXP(ETA(4))                    ; Emax

TVC50 = THETA(8)                          ; typical value of C50
C50 = TVC50 * EXP(ETA(5))                     ; C50

TVBASE = THETA(9)                          ; typical value of baseline serum uric acid
TVBASE = BASECOV*TVBASE                        ; [BASE~SCR+GENOTYPE]
BASE = TVBASE * EXP(ETA(6))                   ; baseline serum uric acid

;----------------ODES--------------------

$DES
DADT(1) = -KA * A(1)                       ; ODE for absorption compartment
DADT(2) = KA * A(1) - KE * A(2)                ; ODE for plasma central compartment

;----------------PREDICTION & ERROR MODEL---------------------

$ERROR
CP = A(2)/V
E = BASE - (EMAX * CP)/(C50 + CP)
IPRED=A(2)/V                           ; prediction for PK units: ng/mL
IF(FLAG.EQ.3) IPRED = E                     ; prediction for PD units: mg/dL
WA=THETA(1)                            ; additive error for PK
IF(FLAG.EQ.3) WA = THETA(10)                   ; additive error for PD
WP=THETA(2)                            ; proportional error for PK
IF(FLAG.EQ.3) WP = THETA(11)                   ; proportional error for PD

W = SQRT(WA**2+(WP*IPRED)**2) ; combined error model
IRES=DV-IPRED
IWRES=IRES/W
Y = IPRED + W*EPS(1)

;----------------INITIAL ESTIMATES---------------------

$THETA
 0.01      FIX                        ; additive error for PK
 0.253467  FIX                            ; proportional error for PK
```

```
 1742.83   FIX                    ; CL (mL/hr)
 57013.5   FIX                    ; V (mL)
 0.77117   FIX                    ; KA (1/hr)
 -0.248189 FIX                     ; [V~GENDER]
 1         FIX                  ; EMAX
 (0,2606.22)                      ; C50 (ng/mL)
 4.36262                         ; BASE (mg/dL)
 0         FIX                  ; additive error for PD
 (0,0.0767211)                     ; proportional error for PD
 (-1,0.358231,5)                   ; [BASE~GENOTYPE]
 (-1,0.258194,5)                   ; [BASE~GENOTYPE]
 (-2.500,0.5096,3.125)                ; [BASE~SCR]

$OMEGA
 0.0556119 FIX ; ETA1 (CL)
 0.034849  FIX ; ETA2 (V)
 0.304355  FIX ; ETA3 (KA)
 0         FIX ; ETA4 (EMAX)
 0         FIX ; ETA5 (C50)
 0.0101588     ; ETA6 (BASE)

$SIGMA
 1  FIX

$ESTIMATION METHOD=1 INTERACTION PRINT=5 SIG=3 MAXEVAL=9999 NOABORT
$COVARIANCE UNCONDITIONAL

$TABLE      ID TIME IPRED DV GENDER GENOTYPE SCR SCRSUB WRES CWRES EVID CMTT MDV
        FILE=sdtab21
        NOPRINT ONEHEADER
```